NAG Fortran Library Routine Document

G05ECF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05ECF sets up the reference vector R for a Poisson distribution with mean t.

2 Specification

```
SUBROUTINE G05ECF(T, R, NR, IFAIL)
INTEGER           NR, IFAIL
real              T, R(NR)
```

3 Description

G05ECF sets up a reference vector for use in G05EYF. Together these routines produce random numbers from the distribution defined by:

$$P(I = i) = \frac{t^i e^{-t}}{i!} \quad \text{if} \quad i = 0, 1, \ldots,$$

$$P(I = i) = 0 \qquad \text{otherwise.}$$

The reference array is found using a recurrence relation if t is less than 50 and by Stirling's formula otherwise.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

Kendall M G and Stuart A (1969) *The Advanced Theory of Statistics (Volume 1)* (3rd Edition) Griffin

5 Parameters

1: T – ***real*** *Input*

On entry: the mean, t, of the distribution.

Constraint: $\mathrm{T} \geq 0$.

2: R(NR) – ***real*** array *Output*

On exit: the reference vector.

3: NR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05ECF is called.

Suggested value: approximately $20 + 20 \times \sqrt{\mathrm{T}}$ (for optimum efficiency in G05EYF).

Constraint: $\mathrm{NR} > (\mathrm{INT}[\mathrm{T} + 7.15\sqrt{\mathrm{T}} + 8.5] - \max(0, \mathrm{INT}[\mathrm{T} - 7.15\sqrt{\mathrm{T}}]) + 4)$.

4: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, T < 0.

IFAIL = 2

On entry, NR is too small (see Section 5).

7 Accuracy

Not applicable.

8 Further Comments

The time taken by the routine increases with NR.

9 Example

The example program sets up a reference for a Poisson distribution with mean 2.7 and then prints the first five pseudo-random numbers generated by G05EYF, after initialisation by G05CBF.

The generator mechanism used is selected by an initial call to G05ZAF.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05ECF Example Program Text
*     Mark 20 Revised. NAG Copyright 2001.
*     .. Parameters ..
      real              T
      INTEGER           NR
      PARAMETER         (T=2.7e0,NR=30)
      INTEGER           NOUT
      PARAMETER         (NOUT=6)
*     .. Local Scalars ..
      INTEGER           I, IFAIL, IX
*     .. Local Arrays ..
      real              R(NR)
*     .. External Functions ..
      INTEGER           G05EYF
      EXTERNAL          G05EYF
*     .. External Subroutines ..
      EXTERNAL          G05CBF, G05ECF, G05ZAF
*     .. Executable Statements ..
      CALL G05ZAF('O')
      WRITE (NOUT,*) 'G05ECF Example Program Results'
      WRITE (NOUT,*)
      CALL G05CBF(0)
```

```
      IFAIL = 0
*
      CALL G05ECF(T,R,NR,IFAIL)
*
      DO 20 I = 1, 5
         IX = G05EYF(R,30)
         WRITE (NOUT,99999) IX
   20 CONTINUE
      STOP
*
99999 FORMAT (1X,I5)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05ECF Example Program Results

     4
     1
     2
     1
     5
```